NAG Library Routine Document

G13FAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G13FAF estimates the parameters of either a standard univariate regression GARCH process, or a univariate regression-type I AGARCH(p,q) process (see Engle and Ng (1993)).

2 Specification

```
SUBROUTINE G13FAF (DIST, YT, X, LDX, NUM, IP, IQ, NREG, MN, ISYM, NPAR,       &
                   THETA, SE, SC, COVR, LDCOVR, HP, ET, HT, LGF, COPTS,       &
                   MAXIT, TOL, WORK, LWORK, IFAIL)
INTEGER            LDX, NUM, IP, IQ, NREG, MN, ISYM, NPAR, LDCOVR, MAXIT,     &
                   LWORK, IFAIL
REAL (KIND=nag_wp) YT(NUM), X(LDX,*), THETA(NPAR), SE(NPAR), SC(NPAR),        &
                   COVR(LDCOVR,NPAR), HP, ET(NUM), HT(NUM), LGF, TOL,         &
                   WORK(LWORK)
LOGICAL            COPTS(2)
CHARACTER(1)       DIST
```

3 Description

A univariate regression-type I AGARCH(p,q) process, with q coefficients α_i, for $i=1,2,\dots,q$, p coefficients β_i, for $i=1,2,\dots,p$, and k linear regression coefficients b_i, for $i=1,2,\dots,k$, can be represented by:

$$y_t = b_o + x_t^{\mathrm{T}} b + \epsilon_t \tag{1}$$

$$h_t = \alpha_0 + \sum_{i=1}^{q} \alpha_i (\epsilon_{t-i} + \gamma)^2 + \sum_{i=1}^{p} \beta_i h_{t-i}, \qquad t = 1,2,\dots,T \tag{2}$$

where $\epsilon_t \mid \psi_{t-1} = N(0,h_t)$ or $\epsilon_t \mid \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t-distribution with df degrees of freedom and variance h_t, T is the number of terms in the sequence, y_t denotes the endogenous variables, x_t the exogenous variables, b_o the regression mean, b the regression coefficients, ϵ_t the residuals, h_t the conditional variance, df the number of degrees of freedom of the Student's t-distribution, and ψ_t the set of all information up to time t.

G13FAF provides an estimate for $\hat{\theta}$, the parameter vector $\theta = \left(b_o, b^{\mathrm{T}}, \omega^{\mathrm{T}}\right)$ where $b^{\mathrm{T}} = (b_1,\dots,b_k)$, $\omega^{\mathrm{T}} = \left(\alpha_0, \alpha_1, \dots, \alpha_q, \beta_1, \dots, \beta_p, \gamma\right)$ when DIST = 'N' and $\omega^{\mathrm{T}} = \left(\alpha_0, \alpha_1, \dots, \alpha_q, \beta_1, \dots, \beta_p, \gamma, df\right)$ when DIST = 'T'.

ISYM, MN and NREG can be used to simplify the GARCH(p,q) expression in (1) as follows:

No Regression and No Mean

$y_t = \epsilon_t$,

ISYM $= 0$,

MN $= 0$,

NREG $= 0$ and

θ is a $(p+q+1)$ vector when DIST = 'N' and a $(p+q+2)$ vector when DIST = 'T'.

No Regression

$y_t = b_o + \epsilon_t$,

ISYM = 0,

MN = 1,

NREG = 0 and

θ is a $(p + q + 2)$ vector when DIST = 'N' and a $(p + q + 3)$ vector when DIST = 'T'.

Note: if the $y_t = \mu + \epsilon_t$, where μ is known (not to be estimated by G13FAF) then (1) can be written as $y_t^\mu = \epsilon_t$, where $y_t^\mu = y_t - \mu$. This corresponds to the case **No Regression and No Mean**, with y_t replaced by $y_t - \mu$.

No Mean

$y_t = x_t^{\mathrm{T}} b + \epsilon_t$,

ISYM = 0,

MN = 0,

NREG = k and

θ is a $(p + q + k + 1)$ vector when DIST = 'N' and a $(p + q + k + 2)$ vector when DIST = 'T'.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and testing the impact of news on volatility *Journal of Finance* **48** 1749–1777

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER(1) *Input*

On entry: the type of distribution to use for e_t.

DIST = 'N'
A Normal distribution is used.

DIST = 'T'
A Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: YT(NUM) – REAL (KIND=nag_wp) array *Input*

On entry: the sequence of observations, y_t, for $t = 1, 2, \ldots, T$.

3: X(LDX,*) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array X must be at least NREG.

On entry: row t of X must contain the time dependent exogenous vector x_t, where $x_t^{\mathrm{T}} = \left(x_t^1, \ldots, x_t^k\right)$, for $t = 1, 2, \ldots, T$.

4: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G13FAF is called.

Constraint: $\text{LDX} \geq \text{NUM}$.

5: NUM – INTEGER *Input*

On entry: T, the number of terms in the sequence.

Constraints:

$\text{NUM} \geq \max(\text{IP}, \text{IQ})$;
$\text{NUM} \geq \text{NREG} + \text{MN}$.

6: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, 2, \ldots, p$.

Constraint: $\text{IP} \geq 0$ (see also NPAR).

7: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, 2, \ldots, q$.

Constraint: $\text{IQ} \geq 1$ (see also NPAR).

8: NREG – INTEGER *Input*

On entry: k, the number of regression coefficients.

Constraint: $\text{NREG} \geq 0$ (see also NPAR).

9: MN – INTEGER *Input*

On entry: if $\text{MN} = 1$, the mean term b_0 will be included in the model.

Constraint: $\text{MN} = 0$ or 1.

10: ISYM – INTEGER *Input*

On entry: if $\text{ISYM} = 1$, the asymmetry term γ will be included in the model.

Constraint: $\text{ISYM} = 0$ or 1.

11: NPAR – INTEGER *Input*

On entry: the number of parameters to be included in the model. $\text{NPAR} = 1 + \text{IQ} + \text{IP} + \text{ISYM} + \text{MN} + \text{NREG}$ when DIST = 'N', and $\text{NPAR} = 2 + \text{IQ} + \text{IP} + \text{ISYM} + \text{MN} + \text{NREG}$ when DIST = 'T'.

Constraint: $\text{NPAR} < 20$.

12: THETA(NPAR) – REAL (KIND=nag_wp) array *Input/Output*

On entry: the initial parameter estimates for the vector θ.

The first element must contain the coefficient α_o and the next IQ elements must contain the coefficients α_i, for $i = 1, 2, \ldots, q$.

The next IP elements must contain the coefficients β_j, for $j = 1, 2, \ldots, p$.

If $\text{ISYM} = 1$, the next element must contain the asymmetry parameter γ.

If DIST = 'T', the next element must contain *df*, the number of degrees of freedom of the Student's t-distribution.

If $\text{MN} = 1$, the next element must contain the mean term b_o.

If COPTS(2) = .FALSE., the remaining NREG elements are taken as initial estimates of the linear regression coefficients b_i, for $i = 1, 2, \ldots, k$.

On exit: the estimated values $\hat{\theta}$ for the vector θ.

The first element contains the coefficient α_o, the next IQ elements contain the coefficients α_i, for $i = 1, 2, \ldots, q$.

The next IP elements are the coefficients β_j, for $j = 1, 2, \ldots, p$.

If ISYM = 1, the next element contains the estimate for the asymmetry parameter γ.

If DIST = 'T', the next element contains an estimate for *df*, the number of degrees of freedom of the Student's t-distribution.

If MN = 1, the next element contains an estimate for the mean term b_o.

The final NREG elements are the estimated linear regression coefficients b_i, for $i = 1, 2, \ldots, k$.

13: SE(NPAR) – REAL (KIND=nag_wp) array *Output*

On exit: the standard errors for $\hat{\theta}$.

The first element contains the standard error for α_o. The next IQ elements contain the standard errors for α_i, for $i = 1, 2, \ldots, q$. The next IP elements are the standard errors for β_j, for $j = 1, 2, \ldots, p$.

If ISYM = 1, the next element contains the standard error for γ.

If DIST = 'T', the next element contains the standard error for *df*, the number of degrees of freedom of the Student's t-distribution.

If MN = 1, the next element contains the standard error for b_o.

The final NREG elements are the standard errors for b_j, for $j = 1, 2, \ldots, k$.

14: SC(NPAR) – REAL (KIND=nag_wp) array *Output*

On exit: the scores for $\hat{\theta}$.

The first element contains the score for α_o.

The next IQ elements contain the score for α_i, for $i = 1, 2, \ldots, q$.

The next IP elements are the scores for β_j, for $j = 1, 2, \ldots, p$.

If ISYM = 1, the next element contains the score for γ.

If DIST = 'T', the next element contains the score for *df*, the number of degrees of freedom of the Student's t-distribution.

If MN = 1, the next element contains the score for b_o.

The final NREG elements are the scores for b_j, for $j = 1, 2, \ldots, k$.

15: COVR(LDCOVR,NPAR) – REAL (KIND=nag_wp) array *Output*

On exit: the covariance matrix of the parameter estimates $\hat{\theta}$, that is the inverse of the Fisher Information Matrix.

16: LDCOVR – INTEGER *Input*

On entry: the first dimension of the array COVR as declared in the (sub)program from which G13FAF is called.

Constraint: LDCOVR $\geq$ NPAR.

17: HP – REAL (KIND=nag_wp) *Input/Output*

On entry: if COPTS(2) = .FALSE., HP is the value to be used for the pre-observed conditional variance; otherwise HP is not referenced.

On exit: if COPTS(2) = .TRUE., HP is the estimated value of the pre-observed conditional variance.

18: ET(NUM) – REAL (KIND=nag_wp) array *Output*

On exit: the estimated residuals, ϵ_t, for $t = 1, 2, \dots, T$.

19: HT(NUM) – REAL (KIND=nag_wp) array *Output*

On exit: the estimated conditional variances, h_t, for $t = 1, 2, \dots, T$.

20: LGF – REAL (KIND=nag_wp) *Output*

On exit: the value of the log-likelihood function at $\hat{\theta}$.

21: COPTS(2) – LOGICAL array *Input*

On entry: the options to be used by G13FAF.

COPTS(1) = .TRUE.
: Stationary conditions are enforced, otherwise they are not.

COPTS(2) = .TRUE.
: The routine provides initial parameter estimates of the regression terms, otherwise these are to be provided by you.

22: MAXIT – INTEGER *Input*

On entry: the maximum number of iterations to be used by the optimization routine when estimating the GARCH(p, q) parameters. If MAXIT is set to 0, the standard errors, score vector and variance-covariance are calculated for the input value of θ in THETA; however the value of θ is not updated.

Constraint: MAXIT ≥ 0.

23: TOL – REAL (KIND=nag_wp) *Input*

On entry: the tolerance to be used by the optimization routine when estimating the GARCH(p, q) parameters.

24: WORK(LWORK) – REAL (KIND=nag_wp) array *Workspace*
25: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which G13FAF is called.

Constraint: LWORK $\geq$ (NREG + 3) $\times$ NUM + NPAR + 403.

26: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: G13FAF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, NREG < 0,
or MN > 1,
or MN < 0,
or ISYM > 1,
or ISYM < 0,
or IQ < 1,
or IP < 0,
or NPAR ≥ 20,
or NPAR has an invalid value,
or LDCOVR $<$ NPAR,
or LDX $<$ NUM,
or DIST $\neq$ 'N',
or DIST $\neq$ 'T',
or MAXIT < 0,
or NUM $< \max(\mathrm{IP}, \mathrm{IQ})$,
or NUM $<$ NREG $+$ MN.

IFAIL $= 2$

On entry, LWORK $< (\mathrm{NREG} + 3) \times \mathrm{NUM} + \mathrm{NPAR} + 403$.

IFAIL $= 3$

The matrix X is not full rank.

IFAIL $= 4$

The information matrix is not positive definite.

IFAIL $= 5$

The maximum number of iterations has been reached.

IFAIL $= 6$

The log-likelihood cannot be optimized any further.

IFAIL $= 7$

No feasible model parameters could be found.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example fits a GARCH(1, 1) model with Student's t-distributed residuals to some simulated data.

The process parameter estimates, $\hat{\theta}$, are obtained using G13FAF, and a four step ahead volatility estimate is computed using G13FBF.

The data was simulated using G05PDF.

9.1 Program Text

```
    Program g13fafe

!     G13FAF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g13faf, g13fbf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                :: gamma, hp, lgf, tol
      Integer                           :: i, ifail, ip, iq, isym, l, ldcovr,   &
                                           ldx, lwork, maxit, mn, npar, nreg,   &
                                           nt, num, pgamma
      Logical                           :: tdist
      Character (1)                     :: dist
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable   :: covr(:,:), et(:), fht(:), ht(:),     &
                                           sc(:), se(:), theta(:), work(:),     &
                                           x(:,:), yt(:)
      Logical                           :: copts(2)
!     .. Executable Statements ..
      Write (nout,*) 'G13FAF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Read in the problem size
      Read (nin,*) num, mn, nreg

      ldx = num
      Allocate (yt(num),x(ldx,nreg))

!     Read in the series
      Read (nin,*) yt(1:num)

!     Read in the exogenous variables
      If (nreg>0) Then
        Read (nin,*,Iostat=ifail)(x(i,1:nreg),i=1,num)
      End If

!     Read in details of the model to fit
      Read (nin,*) dist, ip, iq, isym

!     Read in control parameters
      Read (nin,*) copts(1:2), maxit, tol

!     Calculate NPAR
      npar = 1 + iq + ip + isym + mn + nreg
      If (dist=='T' .Or. dist=='t') Then
        npar = npar + 1
        tdist = .True.
      Else
        tdist = .False.
      End If
```

```
      ldcovr = npar
      lwork = (nreg+3)*num + npar + 403
      Allocate (theta(npar),se(npar),sc(npar),covr(ldcovr,npar),et(num), &
        ht(num),work(lwork))

!     Read in initial values
!     alpha_0
      Read (nin,*) theta(1)
      l = 2
!     alpha_i
      If (iq>0) Then
        Read (nin,*) theta(l:(l+iq-1))
        l = l + iq
      End If
!     beta_i
      If (ip>0) Then
        Read (nin,*) theta(l:(l+ip-1))
        l = l + ip
      End If
!     gamma
      If (isym==1) Then
        Read (nin,*) theta(l)
        pgamma = l
        l = l + 1
      End If
!     degrees of freedom
      If (tdist) Then
        Read (nin,*) theta(l)
        l = l + 1
      End If
!     mean
      If (mn==1) Then
        Read (nin,*) theta(l)
        l = l + 1
      End If
!     Regression parameters and pre-observed conditional variance
      If (.Not. copts(2)) Then
        Read (nin,*) theta(l:(l+nreg-1))
        Read (nin,*) hp
      End If

!     Fit the GARCH model
      ifail = -1
      Call g13faf(dist,yt,x,ldx,num,ip,iq,nreg,mn,isym,npar,theta,se,sc,covr, &
        ldcovr,hp,et,ht,lgf,copts,maxit,tol,work,lwork,ifail)
      If (ifail/=0) Then
        If (ifail/=5 .And. ifail/=6) Then
          Go To 100
        End If
      End If

!     Read in forecast horizon
      Read (nin,*) nt

      Allocate (fht(nt))

!     Extract the estimate of the asymmetry parameter from theta
      If (isym==1) Then
        gamma = theta(pgamma)
      Else
        gamma = 0.0E0_nag_wp
      End If

!     Calculate the volatility forecast
      ifail = 0
      Call g13fbf(num,nt,ip,iq,theta,gamma,fht,ht,et,ifail)

!     Output the results
      Write (nout,*) '               Parameter        Standard'
      Write (nout,*) '               estimates         errors'
```

```
!     Output the coefficient alpha_0
      Write (nout,99999) 'Alpha', 0, theta(1), se(1)
      l = 2
!     Output the coefficients alpha_i
      If (iq>0) Then
        Write (nout,99999)('Alpha',i-1,theta(i),se(i),i=l,l+iq-1)
        l = l + iq
      End If
      Write (nout,*)
!     Output the coefficients beta_j
      If (ip>0) Then
        Write (nout,99999)(' Beta',i-l+1,theta(i),se(i),i=l,l+ip-1)
        l = l + ip
        Write (nout,*)
      End If
!     Output the estimated asymmetry parameter, gamma
      If (isym==1) Then
        Write (nout,99998) ' Gamma', theta(l), se(l)
        Write (nout,*)
        l = l + 1
      End If
!     Output the estimated degrees of freedom, df
      If (dist=='T') Then
        Write (nout,99998) '    DF', theta(l), se(l)
        Write (nout,*)
        l = l + 1
      End If
!     Output the estimated mean term, b_0
      If (mn==1) Then
        Write (nout,99999) '    B', 0, theta(l), se(l)
        l = l + 1
      End If
!     Output the estimated linear regression coefficients, b_i
      If (nreg>0) Then
        Write (nout,99999)('    B',i-l+1,theta(i),se(i),i=l,l+nreg-1)
      End If
      Write (nout,*)

!     Display the volatility forecast
      Write (nout,99997) 'Volatility forecast = ', fht(nt)
      Write (nout,*)

100   Continue

99999 Format (1X,A,I0,1X,2F16.2)
99998 Format (1X,A,1X,2F16.2)
99997 Format (1X,A,F12.2)
    End Program g13fafe
```

9.2 Program Data

```
 G13FAF Example Program Data
100 1 2                                     :: NUM,MN,NREG
 9.04  9.49  9.12  9.23  9.35
 9.09  9.75  9.23  8.76  9.17
 9.20  9.64  8.74  9.23  9.42
 9.70  9.55 10.00  9.18  9.77
 9.80  9.56  9.28  9.68  9.51
 9.51  8.97  9.30  9.52  9.41
 9.53  9.75  9.72  9.38  9.28
 9.42  9.74  9.75  9.60  9.90
 9.06  9.92  9.21  9.57  9.42
 8.65  8.85  9.61 10.77 10.19
10.47 10.10 10.21  9.96  9.66
 9.79 10.30  9.68 10.08 10.38
 9.69  9.02  9.89 10.46 10.47
 9.99  9.76  9.78  9.62 10.43
10.42  9.95  9.95  9.70 10.24
 9.78  9.98  8.73 10.23  9.10
10.27  9.85 10.44 10.30 10.08
```

```
10.20 10.14  9.89  9.90 11.33
 9.71  9.40  9.97 10.92  9.76
10.16 10.43  9.60 10.29 10.03                :: End of Y
 0.12  2.40      0.12  2.40
 0.13  2.40      0.14  2.40
 0.14  2.40      0.15  2.40
 0.16  2.40      0.16  2.40
 0.17  2.40      0.18  2.41
 0.19  2.41      0.19  2.41
 0.20  2.41      0.21  2.41
 0.21  2.41      0.22  2.41
 0.23  2.41      0.23  2.41
 0.24  2.41      0.25  2.42
 0.25  2.42      0.26  2.42
 0.26  2.42      0.27  2.42
 0.28  2.42      0.28  2.42
 0.29  2.42      0.30  2.42
 0.30  2.42      0.31  2.43
 0.32  2.43      0.32  2.43
 0.33  2.43      0.33  2.43
 0.34  2.43      0.35  2.43
 0.35  2.43      0.36  2.43
 0.37  2.43      0.37  2.44
 0.38  2.44      0.38  2.44
 0.39  2.44      0.39  2.44
 0.40  2.44      0.41  2.44
 0.41  2.44      0.42  2.44
 0.42  2.44      0.43  2.45
 0.43  2.45      0.44  2.45
 0.45  2.45      0.45  2.45
 0.46  2.45      0.46  2.45
 0.47  2.45      0.47  2.45
 0.48  2.45      0.48  2.46
 0.49  2.46      0.49  2.46
 0.50  2.46      0.50  2.46
 0.51  2.46      0.51  2.46
 0.52  2.46      0.52  2.46
 0.53  2.46      0.53  2.47
 0.54  2.47      0.54  2.47
 0.54  2.47      0.55  2.47
 0.55  2.47      0.56  2.47
 0.56  2.47      0.57  2.47
 0.57  2.47      0.57  2.48
 0.58  2.48      0.58  2.48
 0.59  2.48      0.59  2.48
 0.59  2.48      0.60  2.48
 0.60  2.48      0.61  2.48
 0.61  2.48      0.61  2.49
 0.62  2.49      0.62  2.49
 0.62  2.49      0.63  2.49
 0.63  2.49      0.63  2.49
 0.64  2.49      0.64  2.49
 0.64  2.49      0.64  2.50                  :: End of X
 'T' 1 1 1                                   :: DIST,IP,IQ,ISYM
 T T 200 0.00001                             :: COPTS,MAXIT,TOL
 0.05                                        :: ALPHA_0
 0.10                                        :: ALPHA_I
 0.15                                        :: BETA_I
-0.10                                        :: GAMMA
 2.60                                        :: DF
 1.50                                        :: MEAN
 4                                           :: NT
```

9.3 Program Results

```
 G13FAF Example Program Results

               Parameter        Standard
               estimates         errors
 Alpha0            0.00            0.06
```

```
Alpha1            0.11          0.13

 Beta1            0.66          0.23

 Gamma           -0.62          0.62

    DF            6.25          4.70

    B0            3.85         24.11
    B1            1.48          1.82
    B2            2.15         10.16

Volatility forecast =      0.09
```